https://github.com/t-o-k/Maxima-bezier/rational_bezier_curves_2d.wxmx

Resources:

"Drawing Circles with Rational Quadratic Bezier Curves"
https://ctan.uib.no/macros/latex/contrib/lapdf/rcircle.pdf

"Graphics with MAXIMA"
http://www.austromath.at/daten/maxima/zusatz/Graphics_with_Maxima.pdf

(%i1)
```
kill(all)$
```

(%i2)
```
load("draw")$
load("bezier")$
```

(%i3)
```
tau: 2*%pi$
```

(%i4)
```
no_of_segments: 3$
```

(%i5)
```
angle: tau/no_of_segments/2;
```

(angle) $\frac{\pi}{3}$

(%i6)
```
r2: r1/cos(angle);
```

(r2) $2\,r1$

(%i7)
```
points_evn_x: r1*cos(angle*[ 0, 2, 4 ]);
```

(points_evn_x) $\left[r1,-\frac{r1}{2},-\frac{r1}{2}\right]$

(%i8)
```
points_evn_y: r1*sin(angle*[ 0, 2, 4 ]);
```

(points_evn_y) $\left[0,\frac{\sqrt{3}\,r1}{2},-\frac{\sqrt{3}\,r1}{2}\right]$

(%i9)
```
points_odd_x: r2*cos(angle*[ 1, 3, 5 ]);
```

(points_odd_x) $\left[r1,-2\,r1,r1\right]$

(%i10)
```
points_odd_y: r2*sin(angle*[ 1, 3, 5 ]);
```

(points_odd_y) $\left[\sqrt{3}\,r1,0,-\sqrt{3}\,r1\right]$

(%i12)
```
points1_x: [ points_evn_x[1], points_odd_x[1], points_evn_x[2] ]$
points1_y: [ points_evn_y[1], points_odd_y[1], points_evn_y[2] ]$
```

(%i14)
```
points2_x: [ points_evn_x[2], points_odd_x[2], points_evn_x[3] ]$
points2_y: [ points_evn_y[2], points_odd_y[2], points_evn_y[3] ]$
```

(%i16)
```
points3_x: [ points_evn_x[3], points_odd_x[3], points_evn_x[1] ]$
points3_y: [ points_evn_y[3], points_odd_y[3], points_evn_y[1] ]$
```

(%i17)
```
weights: matrix([ 1, cos(angle), 1 ]);
```

(weights)
$$\begin{pmatrix} 1 & \frac{1}{2} & 1 \end{pmatrix}$$

(%i18)
```
define(f1_x(s), rational_bezier_function_1a(matrix(points1_x), weights, s));
```

(%o18)
$$f1_x(s) := \frac{-\frac{r1\,s^2}{2} + r1\,(1-s)\,s + r1\,(1-s)^2}{s^2 + (1-s)\,s + (1-s)^2}$$

(%i19)
```
define(f1_y(s), rational_bezier_function_1a(matrix(points1_y), weights, s));
```

(%o19)
$$f1_y(s) := \frac{\frac{\sqrt{3}\,r1\,s^2}{2} + \sqrt{3}\,r1\,(1-s)\,s}{s^2 + (1-s)\,s + (1-s)^2}$$

(%i20)
```
define(f2_x(s), rational_bezier_function_1a(matrix(points2_x), weights, s));
```

(%o20)
$$f2_x(s) := \frac{-\frac{r1\,s^2}{2} - 2\,r1\,(1-s)\,s - \frac{r1\,(1-s)^2}{2}}{s^2 + (1-s)\,s + (1-s)^2}$$

(%i21)
```
define(f2_y(s), rational_bezier_function_1a(matrix(points2_y), weights, s));
```

(%o21)
$$f2_y(s) := \frac{\frac{\sqrt{3}\,r1\,(1-s)^2}{2} - \frac{\sqrt{3}\,r1\,s^2}{2}}{s^2 + (1-s)\,s + (1-s)^2}$$

(%i22)
```
define(f3_x(s), rational_bezier_function_1a(matrix(points3_x), weights, s));
```

(%o22)
$$f3_x(s) := \frac{r1\,s^2 + r1\,(1-s)\,s - \frac{r1\,(1-s)^2}{2}}{s^2 + (1-s)\,s + (1-s)^2}$$

(%i23) define(f3_y(s), rational_bezier_function_1a(matrix(points3_y), weights, s));

(%o23) $$f3_y(s):=\frac{-\sqrt{3}\,r1\,(1-s)\,s-\frac{\sqrt{3}\,r1\,(1-s)^2}{2}}{s^2+(1-s)\,s+(1-s)^2}$$

(%i24) r1: 2$

```
(%i25) wxdraw2d(
         title = "Circle made with 3 rational Bezier curves",
         proportional_axes = xy,
         xrange = [ -6, +5 ],
         yrange = [ -5, +5 ],
         line_width = 4,
         color = red,
         parametric(f1_x(s), f1_y(s), s, 0, 1),
         color = green,
         parametric(f2_x(s), f2_y(s), s, 0, 1),
         color = blue,
         parametric(f3_x(s), f3_y(s), s, 0, 1),
         line_width = 2,
         line_type = dashes,
         point_type = none,
         points_joined = true,
         color = red,
         points(''points1_x, ''points1_y),
         color = green,
         points(''points2_x, ''points2_y),
         color = blue,
         points(''points3_x, ''points3_y),
         point_size = 2,
         point_type = filled_circle,
         points_joined = false,
         color = black,
         points(''points_evn_x, ''points_evn_y),
         color = gray,
         points(''points_odd_x, ''points_odd_y)
       );
```

(%t25)



(%o25)