

https://github.com/t-o-k/Maxima-bezier/bezier_surface_with_control_grid_3d.wmx

Copyright (c) 2020 Tor Olav Kristensen, <http://subcube.com>

Use of this source code is governed by the GNU Lesser General Public License version 3, which can be found in the LICENSE file.

```
(%i1) kill(all)$
```

```
(%i2) load("draw")$  
load("bezier")$
```

```
(%i3) combine_xyz(xx, yy, zz) := map(lambda([ x, y, z ], [ x, y, z ]), xx, yy, zz)$
```

```
(%i4) points2a_x:  
[  
  [ 0.0, 1.0, 2.0, 3.0 ],  
  [ 0.0, 1.0, 2.0, 4.0 ],  
  [ 0.0, 1.0, 2.0, 2.5 ],  
  [ 0.0, 1.0, 2.0, 3.0 ]  
]  
$
```

```
(%i5) points2a_y:  
[  
  [ 0.0, 0.0, 1.0, 0.0 ],  
  [ 1.0, 1.0, 2.0, 1.0 ],  
  [ 2.0, 2.0, 3.0, 2.0 ],  
  [ 3.0, 3.0, 5.0, 3.0 ]  
]  
$
```

```
(%i6) points2a_z:  
[  
  [ 2.0, 0.0, 0.0, -3.0 ],  
  [ -2.0, -3.0, -2.0, 3.0 ],  
  [ 0.0, -4.0, 0.0, 2.0 ],  
  [ 2.0, 0.0, 0.0, -3.0 ]  
]  
$
```

```
(%i9) p_x: apply(matrix, points2a_x)$  
p_y: apply(matrix, points2a_y)$  
p_z: apply(matrix, points2a_z)$
```

```
(%i12) define(s_x(u, v), bezier_function_2a(p_x, u, v))$
define(s_y(u, v), bezier_function_2a(p_y, u, v))$
define(s_z(u, v), bezier_function_2a(p_z, u, v))$

(%i13) expand(s_x(u, v));
(%o13)  $4.5 u^3 v^3 - 7.5 u^3 v^2 + 3.0 u^3 v + 3.0 u^3$ 

(%i14) expand(s_y(u, v));
(%o14)  $-3.0 u^3 v^3 + 3.0 u^2 v^3 + 3.0 v^3 - 3.0 u^3 + 3.0 u^2$ 

(%i15) expand(s_z(u, v));
(%o15)  $36.0 u^3 v^3 - 54.0 u^2 v^3 + 27.0 u v^3 - 6.0 v^3 - 57.0 u^3 v^2 +$ 
 $54.0 u^2 v^2 - 36.0 u v^2 + 18.0 v^2 + 21.0 u^3 v + 9.0 u v - 12.0 v - 5.0 u^3$ 
 $+ 6.0 u^2 - 6.0 u + 2.0$ 


(%i16) surface:
parametric_surface(
  s_x(u, v),
  s_y(u, v),
  s_z(u, v),
  u, 0, 1,
  v, 0, 1
)
$

(%i18) points2a_xyz:
map(
  combine_xyz,
  points2a_x,
  points2a_y,
  points2a_z
)
$
control_grid: apply(mesh, points2a_xyz)$

(%i20) points1a_xyz:
combine_xyz(
  flatten(points2a_x),
  flatten(points2a_y),
  flatten(points2a_z)
)
$
control_points: points(points1a_xyz)$
```

```
(%i21) wxdraw3d(
    title = "Bezier surface with control grid",
    proportional_axes = xyz,
    xu_grid = 20,
    yv_grid = 20,
    view = [ 6, 331 ],
    color = black,
    line_width = 1,
    wired_surface = true,
    surface,
    color = blue,
    line_width = 2,
    control_grid,
    color = red,
    point_type = filled_circle,
    point_size = 1,
    control_points
);
```

(%t21)



(%o21)